Pairs, Lists, Trees

Scheme supports Pairs

cons : T_1, T_2 → T_1 × T_2

types of pairs

car : T_1 × T_2 → T_1
cdr : T_1 × T_2 → T_2

Pairing Contract

         (car (cons ⟨v_1⟩ ⟨v_2⟩)) = ⟨v_1⟩
         (cdr (cons ⟨v_1⟩ ⟨v_2⟩)) = ⟨v_2⟩

(pair? ⟨p⟩) =⇒

{

#t if ⟨p⟩ is a pair (a cons-cell);
#f otherwise

Three Descriptions of Pairs

Input Expression

(cons 1 2)

Box and Pointer

 1
num

          2
         num

? ?

Output Representation

(1 . 2)

dotted pair notation

Definition:

1. The “empty list”, nil, is a list of type T , for any T .

2. If ⟨L⟩ is a list (of elements of) type T , and ⟨v⟩ is a value of type T , then

(cons ⟨v⟩ ⟨L⟩)

is a list (of elements of) type T .

Can be expressed with a Type Equation: List(T ) = {nil} ⊎ (T × List(T ))

The List Contract

(null? nil) =⇒ #t
(null? ⟨v⟩) =⇒ #f, if ⟨v⟩ ≠ nil

  (car (cons ⟨v⟩ ⟨L⟩)) =⇒ ⟨v⟩
  (cdr (cons ⟨v⟩ ⟨L⟩)) =⇒ ⟨L⟩

Same contract as pairing:

In Scheme, non-null Lists are a Special case of Pairs.

Scheme supports lists

(list ⟨v_0⟩ · · · ⟨v_n⟩) =
    (cons ⟨v_0⟩ (cons . . . (cons ⟨v_n⟩ nil) . . . )

Input Expression:

(define L (list 1 (+ 3 4) (list #t #f)))

Box & Pointer:

? - ?

         1
        num

                    7
                   num

                               #t
                              bool

                                          #f
                                         bool

L - - -

? ? ?

Output Representation:

(1 7 (#t #f))

Scheme supports lists

’() =⇒ the empty list
(length (list 1 2 4)) =⇒ 3
(append (list 3 4) (list 6 8))

=⇒ (3 4 6 8)

        output representation

(list-ref (list ⟨v_0⟩ ⟨v_1⟩ ⟨v_2⟩ · · · ) k) =⇒ ⟨v_k⟩
(list? (list 3 (lambda (x) x))) =⇒ #t
(list? (cons 1 2)) =⇒ #f

These list ops are “built-in” but could all be defined using the list primitives:

pair? null? car cdr cons nil

For example,

(define (length l)
  (if (null? l) 0 (+ 1 (length (cdr l)))))
(define (list-ref l n)
  (if (zero? n)
      (car l)
      (list-ref (cdr l) (- n 1))))
(define (list? l)
  (or (null? l)
      (and (pair? l) (list? (cdr l)))))
length, list? take O(n) steps on a list of length n.
(list-ref l k) takes O(k) steps.

Box and Pointer gives more information than the output representation of a list

Input Output

(list (list 0 1)
      (list 2 1)) ((0 1) (2 1))

(let ((l1 (list 1)))
  (list (cons 0 l1)
        (cons 2 l1)))

                                     ((0 1) (2 1))

-

 0
num

            1
           num

                           2
                          num

                                      1
                                     num

? - ? ? ? ? ? - -

-

 0
num

            1
           num

                           2
                          num

? - ? ? ? ? - ?

List Abstractions

(map ⟨proc⟩ (list ⟨v_0⟩ ⟨v_1⟩ · · · ⟨v_n⟩))

=⇒

        (list (⟨proc⟩ ⟨v_0⟩)
              (⟨proc⟩ ⟨v_1⟩)
                    .

              (⟨proc⟩ ⟨v_n⟩))

(map square (list 1 2 3 4))

=⇒ (1 4 9 16)

(map list (list 1 2 3 4))

=⇒ ((1) (2) (3) (4))

Filter out “good” elements

(filter odd? (list 1 2 3 4 5))

=⇒ (1 3 5)

(filter even? (list 1 2 3 4 5))

=⇒ (2 4)

(define (filter pred lst)
  (cond ((null? lst) nil)
        ((pred (car lst))
         (cons (car lst)
               (filter pred (cdr lst))))
        (else (filter pred (cdr lst)))))

Enumerate Integers

(define (integers-between low high)
  (if (> low high)
      nil
      (cons low (integers-between (+ low 1)
                                  high))))

(integers-between 5 9) -> (5 6 7 8 9)

Examples

(map fib (integers-between 10 20))) -> (55 89 ... 6765)
(filter even? (map fib (integers-between 10 20)))
(map fib (filter even? (integers-between 10 20)))

(accumulate ⟨op⟩ ⟨init⟩ (list ⟨v_0⟩ ⟨v_1⟩ · · · ⟨v_n⟩))

=⇒

    (⟨op⟩ ⟨v_0⟩ (⟨op⟩ ⟨v_1⟩ (⟨op⟩ · · · (⟨op⟩ ⟨v_n⟩ ⟨init⟩))))

(accumulate + 0 (list 1 2 5 6))

=⇒ (+ 1 (+ 2 (+ 5 (+ 6 0)))) =⇒ 14

      (accumulate op init lst)

       lst : List(T )
      init : S
        op : (T, S) → S

accumulate :

( (T, S) → S, S, List(T ) ) → S

Accumulation

(define (accumulate op init lst)
  (if (null? lst)
      init
      (op (car lst)
          (accumulate op init (cdr lst)))))

Sum as Accumulation

(define (addem-up lst)
  (accumulate + 0 lst))

Length as Accumulation

(define (length lst)
  (accumulate (lambda (x y) (+ 1 y))
              0
              lst))

Append as Accumulation

(define (append list1 list2)
  (accumulate cons list2 list1))

Using the List Abstractions

(define (sum-of-odd-squares n)
  (accumulate + 0 ; sum
              (filter odd? ; odd
                      (map square ; squares
                           (integers-between 1 n)))))

 n -_{ints-between} -^{map}square -^{filter}_{odd} -^{accumulate}_+ -_{sum}

Without the Abstractions

(define (sum-of-odd-squares n)
  (if (= n 1)
      1
      (let ((sq (square n)))
        (if (odd? sq)
            (+ sq (sum-of-odd-squares (- n 1)))
            (sum-of-odd-squares (- n 1))))))

“Spaghetti Code”: unclear, easy to have bugs but Faster!

6.001 Sermon: Clearer is better

Optimize later—or leave optimization to the compiler.

Trees

1. A leaf of type T is an element of type T . 2. A Tree of type T ’s is either

(a) a leaf of Type T , or (b) a list of Trees of type T ’s.

Tree(T ) = Leaf(T ) ⊎ (List(Tree(T ))) Leaf(T ) = T

1 4 73

1 3

(define test-tree (list 1
                        (list 4 7 3)
                        (list
                         (list 1 3))))

               test-tree =⇒ (1 (4 7 3) ((1 3)))

output

Scheme Does Not Support Trees (We have to make our own)

For example,

(define (leaf? tr) (not (pair? tr)))
(define (make-leaf el) el) ; el not a pair
(define (make-tree lst-of-trees) lst-of-trees)

(Silly now, but gets serious when trees have labels.) Tree Contract follows from List Contract.

Tree-map : apply procedure to each leaf:

               test-tree =⇒ (1 (4 7 3) ((1 3)))

output

(tree-map square test-tree) =⇒ (1 (16 49 9) ((1 9)))

(define (tree-map proc tr)
  (if (leaf? tr)
      (proc tr)
      (map (lambda (subtree) (tree-map proc subtree))
           tr)))

(define (flatten tree)
  (if (leaf? tree)
      (list tree)
      (accumulate append nil
                  (map flatten tree))))

(define (count-leaves tree)
  (if (leaf? tree)
      1
      (accumulate + 0
                  (map count-leaves tree))))

Abstraction: tree-accumulate (left-to-reader)